

# maple\_ode

2017 年 12 月 1 日

## 1 Table of Contents

- 1 Euler 法, Runge-Kutta 法による微分方程式解法
- 2 Euler 法による落下運動
  - 2.1 重力場中の運動
  - 2.2 Euler 法
  - 2.3 重力場中の運動を Euler 法で解いたら
  - 2.4 空気抵抗がある水滴の落下
- 3 高精度計算
  - 3.1 バネの運動
  - 3.2 2 次の Runge-Kutta の導出
  - 3.3 Runge-Kutta2 次公式
  - 3.4 Runge-Kutta4 次公式
  - 3.5 連立方程式に Runge-Kutta4 次公式を
- 4 RLC 回路の応答
- 5 課題
  - 5.1 雨粒
  - 5.2 大砲
  - 5.3 RLC 回路
  - 5.4 RLC 回路

## 2 Euler 法, Runge-Kutta 法による微分方程式解法

- /Users/bob/Desktop/maple\_ode/maple\_ode.ipynb
- url: [https://github.com/daddygongon/maple\\_ode](https://github.com/daddygongon/maple_ode)
- maple\_ode.ipynb
- cc 2017 by Shigeto R. Nishitani

### 3 Euler 法による落下運動

#### 3.1 重力場中の運動

重力場中のボールの落下を考えて，1 軸で考えた運動方程式を立てます．


$$v = \frac{dx}{dt}$$
$$a = \frac{d^2x}{dt^2}$$

質量を  $m$ ，重力加速度を  $g$  として，働く力が  $F = -mg$  であるとする，ニュートンの運動方程式  $F = ma$  は，

$$-mg = m \frac{d^2x}{dt^2}$$

となります．

## Free gravitation fall



simple\_gravitation\_fall

#### 3.2 Euler 法

1 次の微分方程式の一般形は

$$\frac{dx}{dt} = f(x, t)$$

と書けます。この微分方程式を簡単な近似から求めるオイラー法を示します。  $x(t + \delta t)$  をテイラー級数展開すると、

$$x(t + \delta t) \simeq x(t) + \frac{dx}{dt} \delta t$$

となります。これらを代入すると、計算アルゴリズムはつぎのようになります、

$$x_{i+1} = x_i + f_i \delta t$$

ここで、 $f_i$  は点  $(x_i, t_i)$  における関数の値です。このアルゴリズムを適用して、 $t + \delta t$  の座標  $x_{i+1}$  を一つ前の時間の座標  $x_i$  から導くことができます。これを重力場中の運動方程式に適用します。

### 3.3 重力場中の運動を Euler 法で解いたら

Euler 法は一階の微分方程式に対する定式化をしています。ところが、重力場中の運動は 2 階の微分方程式です。このようなときには媒介変数を導入して 1 次連立方程式に置き直します。

媒介変数として速度  $v$  を使って、2 階の運動方程式

$$\frac{d^2x}{dt^2} = -g$$

が、1 階の連立方程式

$$\begin{aligned} \frac{dv}{dt} &= -g \\ \frac{dx}{dt} &= v \end{aligned}$$

で置き換えられると考えることに相当します。アルゴリズムにすると、

$$\begin{aligned} v_{i+1} &= v_i - g \, dt \\ x_{i+1} &= x_i + v_i \, dt \end{aligned}$$

なる連立方程式を解くことに置き換わります。これを Maple で関数にして、さらに計算結果を表示させてみます。

```
In [ ]: Euler := proc (x0, v0)
    local v1, x1;
    global g, dt;
    v1 := v0 - g * dt;
    x1 := x0 + v0 * dt;
    return x1, v1;
end;
```

Euler は  $x_i, v_i$  を受け取って、先ほど導いた簡単な計算によって、 $v_{i+1}, x_{i+1}$  を順次計算して返します。結果は、

```
In [ ]: dt:=0.1;g:=9.8;
    tt,vv,xx:=[0.0],[0],[10.0];
```


```

for i from 2 to 50 do
  x, v := Euler(xx[i-1],vv[i-1]);
  tt :=[op(tt),tt[-1]+dt];
  xx :=[op(xx),x];
  vv :=[op(vv),v];
end do:

with(plots):
lx:=listplot([tt,xx],color='blue');
lv:=listplot([tt,vv],color='red');
display(lx,lv);

```

位置 ( $x$ ) と速度 ( $v$ ) の時間 ( $t$ ) 変化を表示させています.



euler\_ball\_drop free gravitation fall

時間とともに位置は放物線状に変化し，速度は一定の傾きで増加していく，等加速度運動を再現しています．Euler 法ではこのように非常に簡単な code によって微分方程式で表される現象をシミュレートできることがわかるでしょう．

### 3.4 空気抵抗がある水滴の落下

ball の落下ではわかりにくいですが，より小さな質量の水滴では，速度に比例する空気抵抗が効いてきます．この様子を見ましょう．微分方程式には，


$$F_x = -Cv_x$$

項が付与されます。そうすると運動方程式は

$$m \frac{dv_x}{dt} = -Cv_x - mg$$

となります。これにともなった  $v_x$  の時間変化に対して、今までは単純に重力加速していたのが、 $v_x$  に比例する空気抵抗を記述する項が付与されます。この変化を Euler2 に入れ込むと少しの修正ですが、結果は劇的に変化します。

## gravitation fall with air resistance



gravitation\_fall\_with\_air\_resistance

```
In [ ]: Euler2 := proc (x0, v0)
    local v1, x1;
    global g, dt;
    v1 := v0 + (-cc * v0 - g) * dt;
    x1 := x0 + v0 * dt;
    return x1, v1;
end:


In [ ]: dt:=0.1: g:=9.8: cc:=1:
tt,vv,xx:=[0.0],[0],[10.0];
for i from 2 to 50 do
    x, v := Euler2(xx[i-1],vv[i-1]);
    tt :=[op(tt),tt[-1]+dt];
    xx :=[op(xx),x];
    vv :=[op(vv),v];
```

```

end do:

with(plots):
lx:=listplot([tt,xx],color='blue');
lv:=listplot([tt,vv],color='red');
display(lx,lv);

```



euler\_ball\_drop with air resistance

速度は空気抵抗によって、一定となります。またそれに伴って位置は一定の傾きで落ちていきます。空気抵抗がないときに加速が続く様子とは違った挙動が観察できるでしょう。電車の窓に落ちる雨粒がまっすぐつくのはそのせいです。たぶん...

## 4 高精度計算

### 4.1 バネの運動

今度はバネの運動です。空気抵抗との違いはほんの少しで、

$$F_x = -kx$$

と今度は、位置  $x$  に力が比例することです。そうすると運動方程式は、

$$m \frac{dv_x}{dt} = -kx$$

となります。

連立方程式は


$$\begin{aligned}\frac{dv}{dt} &= -\frac{k}{m}x \\ \frac{dx}{dt} &= v\end{aligned}$$

となるので、アルゴリズムに置き換えると、

$$\begin{aligned}v_{i+1} &= v_i - \frac{k}{m}x_i dt \\ x_{i+1} &= x_i + v_i dt\end{aligned}$$

なる連立方程式を解くことに置き換わります。これを Maple で関数にして、さらに計算結果を表示させてみます。

## Free oscillation mass-spring system



$$\begin{aligned}m \frac{dv}{dt} &= -mg \\ \downarrow \\ m \frac{dv}{dt} &= -(m)k x \\ \frac{v_{i+1} - v_i}{dt} &= -k x \\ v_{i+1} &= v_i + (-k x) dt\end{aligned}$$

free\_oscillation\_mass\_spring\_system

```
In [ ]: Euler3 := proc (x0, v0)
    local v1, x1;
    global k, dx;
    v1 := v0 + (- k * x0 ) * dt;
    x1 := x0 + v0 * dt;
    return x1, v1;
end:

dt:=0.1;k:=0.01;
tt,vv,xx:=[0.0],[0],[10.0];
```


```

for i from 2 to 5000 do
  x, v := Euler3(xx[i-1],vv[i-1]);
  tt :=[op(tt),tt[-1]+dt];
  xx :=[op(xx),x];
  vv :=[op(vv),v];
end do:

with(plots):
lx:=listplot([tt,xx],color='blue');
lv:=listplot([tt,vv],color='red');
display(lx,lv);

```

k はあらかじめ  $m$  で割られて正規化されているとします。これを Euler 法で計算すると次のような結果が得られます。



spring oscillation by Euler

徐々に発散していく様子がわかると思います。本来、摩擦のないバネは定常的に振動します。この発散の原因は、Euler 法の計算誤差が大きいせいです。そこで、より精度の高い Runge-Kutta 法を導入します。

## 4.2 2 次の Runge-Kutta の導出

一般に Runge-Kutta 法と呼ばれる手法は、4 次の古典的 Runge-Kutta 法を指します。導出は意外と面倒なので、2 次の場合のさわりを紹介して、そこからの類推としましょう。  $x, v, t$  などのパラメータ名を変更しますのでご注意ください。



(「ANSI C による数値計算法」堀之内聰一，酒井幸吉，榎園茂，森北 2002, p.133) テイラー展開により， $h^2$  の精度まで展開する．

$$y(x_0 + h) = y(x_0) + y'_0 h + \frac{1}{2!} y''_0 h^2 + O(h^3)$$

この式において， $y'_0 = f(x_0, y_0)$  は既知とする．一方， $y''_0$  は  $f(x_0, y_0)$  から直接的には求められない．したがって，この式の右辺と  $h^2$  の項まで一致する近似値を， $f(x_0, y_0)$  だけを既知として算出する方法を考えよう．

平均値の定理より，

$$\Delta y = y(x_0 + h) - y(x_0) = hy'(x_0 + \theta h), 0 < \theta < 1$$

$y'(x_0 + \theta h)$  の近似値として， $\theta = 0, \theta = 1$  の場合を考えると，

$$\Delta y \simeq hy'(x_0) \quad \text{where } \theta = 0$$

$$\Delta y \simeq hy'(x_0 + h) \quad \text{where } \theta = 1$$

これらの値は単独では  $\Delta y$  に対して  $h^2$  の精度をもつ近似値にならないが，これらの一次結合  $\alpha hy'(x_0) + \beta hy'(x_0 + h)$  を  $\alpha, \beta$  をうまく定めることによって， $\Delta y$  の  $h^2$  の精度をもつ近似値にすることができる．実際，

$$\begin{aligned} \alpha hy'(x_0) + \beta hy'(x_0 + h) &= \alpha hy'(x_0) + \beta h \{y'(x_0) + y''(x_0)h + O(h^2)\} \\ &= (\alpha + \beta)hy'_0 + \beta h^2 y''_0 + O(h^3) \end{aligned}$$

したがって，テイラー展開式と係数を比較して，

$$\begin{aligned} \alpha + \beta &= 1, \beta = \frac{1}{2} \\ \alpha &= \frac{1}{2}, \beta = \frac{1}{2} \end{aligned}$$

となり，

$$\Delta y = \frac{1}{2} hy'(x_0) + \frac{1}{2} hy'(x_0 + h) + O(h^3)$$

いま，

$$k_1 = hy'(x_0) = hf(x_0, y_0)$$

とおこう．上式に代入して，

$$\begin{aligned} hy'(x_0 + h) &= hf(x_0 + h, y(x_0 + h)) \\ &= hf(x_0 + h, y(x_0) + y'(x_0)h + O(h^2)) \\ &= hf(x_0 + h, y_0 + k_1 + O(h^2)) \\ &= hf(x_0 + h, y_0 + k_1) + O(h^2) \end{aligned}$$

したがって，

$$k_2 = hf(x_0 + h, y_0 + k_1)$$

とおけば,

$$\Delta y = \frac{1}{2}k_1 + \frac{1}{2}k_2 + O(h^3)$$

となる. これより,

$$k = \frac{1}{2}(k_1 + k_2), y_1 = y_0 + k$$

とおくと,  $y(x_0 + h) = y_1 + O(h^3)$  となり,  $y_1$  は  $h^2$  の精度の近似値となる.

### 4.3 Runge-Kutta2 次公式

こうして得られた Runge-Kutta の2次の公式を定義すると次の通りです.

微分方程式

$$\frac{dy}{dx} = f(x, y), \text{ where } y(x_0) = y_0$$

の数値解は, 刻み幅を  $h$ ,  $x_n = x_0 + nh$  として, 次の漸化式

$$y_{n+1} = y_n + k(n = 0, 1, 2, \dots)$$

で与えられる. ここに,  $k$  は次で定める.

$$k_1 = hf(x_n, y_n),$$

$$k_2 = hf(x_n + h, y_n + k_1),$$

$$k = \frac{1}{2}(k_1 + k_2)$$

### 4.4 Runge-Kutta4 次公式

2次と同様の考え方で,  $h^4$  の精度を持つ Runge-Kutta 4次公式を作ることができる.

微分方程式

$$\frac{dy}{dx} = f(x, y), \text{ where } y(x_0) = y_0$$

の数値解は, 刻み幅を  $h$ ,  $x_n = x_0 + nh$  として, 次の漸化式

$$y_{n+1} = y_n + k(n = 0, 1, 2, \dots)$$

で与えられる. ここに,  $k$  は次で定める.

$$k_1 = hf(x_n, y_n),$$

$$k_2 = hf(x_n + \frac{h}{2}, y_n + \frac{k_1}{2}),$$

$$k_3 = hf(x_n + \frac{h}{2}, y_n + \frac{k_2}{2}),$$

$$k_4 = hf(x_n + h, y_n + k_3),$$

$$k = \frac{1}{6}(k_1 + 2k_2 + 2k_3 + k_4)$$

## 4.5 連立方程式に Runge-Kutta4 次公式を

連立微分方程式

$$\begin{aligned}\frac{dy}{dx} &= f(x, y, z), \quad \text{where } y(x_0) = y_0 \\ \frac{dz}{dx} &= g(x, y, z), \quad \text{where } z(x_0) = z_0\end{aligned}$$

の数値解は、刻み幅を  $h$ ,  $x_n = x_0 + nh$  として、次の漸化式

$$\begin{aligned}y_{n+1} &= y_n + k \quad (n = 0, 1, 2, \dots) \\ z_{n+1} &= z_n + l \quad (n = 0, 1, 2, \dots)\end{aligned}$$

で与えられる。ここに、 $k, l$  は次で定める。

$$\begin{aligned}k_1 &= hf(x_n, y_n, z_n), & l_1 &= hg(x_n, y_n, z_n), \\ k_2 &= hf(x_n + \frac{h}{2}, y_n + \frac{k_1}{2}, z_n + \frac{l_1}{2}), & l_2 &= hg(x_n + \frac{h}{2}, y_n + \frac{k_1}{2}, z_n + \frac{l_1}{2}), \\ k_3 &= hf(x_n + \frac{h}{2}, y_n + \frac{k_2}{2}, z_n + \frac{l_2}{2}), & l_3 &= hg(x_n + \frac{h}{2}, y_n + \frac{k_2}{2}, z_n + \frac{l_2}{2}), \\ k_4 &= hf(x_n + h, y_n + k_3, z_n + l_3), & l_4 &= hg(x_n + h, y_n + k_3, z_n + l_3), \\ k &= \frac{1}{6}(k_1 + 2k_2 + 2k_3 + k_4), & l &= \frac{1}{6}(l_1 + 2l_2 + 2l_3 + l_4)\end{aligned}$$

```
In [ ]: RungeKutta4 := proc(x_i, y_i, z_i, h)
local k1, l1, k2, l2, k3, l3, k4, l4, kk, ll;

k1:= h * ff(x_i, y_i, z_i);
l1:= h * gg(x_i, y_i, z_i);
k2:= h*ff(x_i+h/2, y_i+k1/2, z_i+l1/2);
l2:= h*gg(x_i+h/2, y_i+k1/2, z_i+l1/2);
k3:= h*ff(x_i+h/2, y_i+k2/2, z_i+l2/2);
l3:= h*gg(x_i+h/2, y_i+k2/2, z_i+l2/2);
k4:= h*ff(x_i+h, y_i+k3, z_i+l3);
l4:= h*gg(x_i+h, y_i+k3, z_i+l3);
kk := 1/6*(k1+2*k2+2*k3+k4);
ll := 1/6*(l1+2*l2+2*l3+l4);
return kk, ll;
end;
```

Runge-Kutta の4次公式をそのまま coding すると上のようになります。これを先ほどのバネ運動の問題に当てはめてみましょう。

先ほど導出した運動方程式の漸化式

$$\begin{aligned}v_{i+1} &= v_i - \frac{k}{m}x_i dt \\ x_{i+1} &= x_i + v_i dt\end{aligned}$$

と Runge-Kutta4 次公式を示した連立微分方程式

$$\frac{dy}{dx} = f(x, y, z), \text{ where } y(x_0) = y_0$$

$$\frac{dz}{dx} = g(x, y, z), \text{ where } z(x_0) = z_0$$

とを比べて、変数の表記の違いと関数  $f, g$  を具体的に書き下します.

4 次の公式	運動方程式
x	t
y	x
z	v
$f(x, y, z)$	$f(t, x, v) = v$
$g(x, y, z)$	$g(t, x, v) = -k x$

この変数の書き換えを吸収する中間関数として Euler3 を書き換えます. RungeKutta4 の仮引数を上の表に従って置き換えて、数値を渡しています. また、関数  $f, g$  も定義しておきます.

```
In [ ]: ff := (t,x,v)-> v;
        gg := (t,x,v)-> -k*x;


ode := proc (x0, v0)
  local v1, x1, kk, ll;
  kk,ll := RungeKutta4(0,x0,v0,dt);
  x1:= x0+kk;
  v1:= v0+ll;
  return x1, v1;
end:

In [ ]: dt:=0.1;k:=0.01;
        tt,vv,xx:= [0.0], [0], [10.0];
        for i from 2 to 5000 do
          x, v := ode(xx[i-1],vv[i-1]);
          tt :=[op(tt),tt[-1]+dt];
          xx :=[op(xx),x];
          vv :=[op(vv),v];
        end do:

        with(plots):
        lx:=listplot([tt,xx],color='blue');
        lv:=listplot([tt,vv],color='red');
```

```
display(lx,lv);
```

これを前と同様に走らせると



spring oscillation by Runge-Kutta 4th order

発散も収束もすることなく、定常的に振動を繰り返していることが見て取れます。

## 5 RLC 回路の応答

電気回路の応答を考えましょう。R(resistor, 抵抗)L(inductor, コイル)C(capacitor, コンデンサー)をそれぞれふたつずつ組み合わせてみていくのが常套手段なんですが、一個ずつだと面倒なので、まずは全部入れた方程式を立てます。そこからパラメータを変えて回路の挙動を観察します。こんなところから進めていけるのが、数値計算の利点です。どんなんでも解けるから。

コンデンサに蓄えられた電荷を  $Q(t)$ , 回路に流れる電流を  $I(t)$  とします。

- 自己インダクタンス  $L$  のコイルにかかる電圧は  $L \frac{dI}{dt}$
- 容量  $C$  のコンデンサにかかる電圧は  $\frac{Q}{C}$
- 抵抗値  $R$  の抵抗に掛かる電圧は  $RI$

となります。コイルにかかる電圧、コンデンサにかかる電圧、抵抗にかかる電圧の和が、この回路にかけた電圧  $V(t)$  であることを使うと、

$$L \frac{dI}{dt} + \frac{Q}{C} + RI = V(t)$$

となります。ここで電流  $I$  とコンデンサの電荷  $Q$  の関係  $I = \frac{dQ}{dt}$  を使うと、

$$L \frac{d^2 Q}{dt^2} + \frac{Q}{C} + R \frac{dQ}{dt} = Vt$$

が得られます.

先ほどの重力系の問題と比べると

$$v \rightarrow i$$

$$x \rightarrow q$$

と置き換えれば良さそうです.


そうするとアルゴリズムは,

$$i_{i+1} = i_i + (V - RR i_i - qc q_i) dt$$

$$q_{i+1} = q_i + i_i dt$$

となりそうです.  $RR$  や  $qc$  には適当に規格化した値をいれます.

## RLC circuit



rlc\_circuit

```
In [ ]: restart;
Euler2 := proc (q0, i0)
    local q1, i1;
    global r,qc,v,dt;
    i1 := i0 + (v - r*i0 -qc*q0) * dt;
    q1 := q0 + i0 * dt;
    return q1, i1;
end;
```


```

In [ ]: dt,r, qc,v := 0.01, 0.5 ,1, 1;
        tt,ii,qq:= [0],[0.0],[0];
        for i from 2 to 2000 do
            q, i2 := Euler2(qq[i-1],ii[i-1]);
            tt :=[op(tt),tt[-1]+dt];
            qq :=[op(qq),q];
            ii :=[op(ii),i2];
        end do:

        with(plots):
        lx:=listplot([tt,qq],color='blue');
        lv:=listplot([tt,ii],color='red');
        display(lx,lv);

```

コンデンサー (C), コンダクター (L) を 1 として, 抵抗値を 0.5 として, 入力電圧を 1V と設定してコンデンサーにたまる電荷と電流の変化を確かめたのが次の図です.



condenser charge in rlc\_circuit

```

In [ ]: dt,r, qc,v := 0.01, 0.5 ,1, 0;
        tt,ii,qq:= [0],[0.0],[1];
        for i from 2 to 2000 do
            q, i2 := Euler2(qq[i-1],ii[i-1]);
            tt :=[op(tt),tt[-1]+dt];
            qq :=[op(qq),q];

```


```

ii :=[op(ii),i2];
end do:

with(plots):
lx:=listplot([tt,qq],color='blue');
lv:=listplot([tt,ii],color='red');
display(lx,lv);

```

入力電圧を 0V, コンデンサーにあらかじめ 1 クーロン貯めたと設定して電荷と電流の時間変化を確かめたのが次の図です.



condenser discharge in rlc\_circuit

## 6 課題

### 6.1 雨粒

空気抵抗を受ける雨粒の速度の変化を自分の言葉で説明しなさい.

### 6.2 大砲

free fall の初期設定を変えて,  $x=0$  から鉛直方向に  $v=1$  で投げ上げたボールの時間変化を示しなさい.



### 6.3 RLC 回路

抵抗が 0 の場合の電荷，電流の変化を確認しなさい．

### 6.4 RLC 回路

抵抗が 0 の場合にも，エネルギー保存が成り立つようにコードを改良しなさい．